

DATEDIF() : Calculer la différence entre deux dates

Catégorie : [Fonctions & Formules](#)

Publié par Mth le 26-07-2009

La fonction DATEDIF() permet de calculer la différence entre deux dates en années, mois et jours.

Cette fonction n'est pas documentée dans l'aide Excel, mais elle s'avère bien pratique pour effectuer des calculs sur des âges ou des dates d'échéance.

Syntaxe : DATEDIF (Date1; Date2;Intervalle)

- Date 1 est la date de Départ
- Date 2 est la date de Fin
- Intervalle peut prendre différentes valeurs:
 - "y" : différence en années
 - "m" : différence en mois
 - "d" : différence en jours
 - "ym" : différence en mois, une fois les années soustraites
 - "yd" : différence en jours, une fois les années soustraites
 - "md" : différence en jours, une fois les années et les mois soustraits



à noter : Date2 doit toujours être supérieure ou égale à Date1

Exemple

Pour illustrer les possibilités de la fonction DATEDIF, prenons l'exemple de 3 chantiers dont nous connaissons les dates de démarrage et de fin, et pour lesquels nous souhaitons connaître la durée :

	B	C	D	E	F	G
	Chantier	Date de démarrage	Date de fin	Nombre d'années	Nombre de mois	Nombre de jours
29						
30	Chantier 1	13/05/2003	11/07/2008	5	61	1 886
31	Chantier 2	03/07/2008	04/08/2009	1	13	397
32	Chantier 3	02/07/2008	03/09/2008	0	2	63
33						

=DATEDIF(C32;D32;"y")

=DATEDIF(C32;D32;"m")

=DATEDIF(C32;D32;"d")

Pour le Chantier 3, les formules inscrites en ligne 32 indiquent qu'il a duré moins d'un an, ou 2 mois, ou encore 63 jours.



Nous remarquons à cette occasion que la fonction DATEDIF renvoie des valeurs entières. Si le chantier 3 a duré 2 mois, la fonction DATEDIF(C32 ;D32 ; "y") n'inscrit pas 0.1667 ans mais bien 0, car il y a moins d'une année complète.

Les trois autres intervalles disponibles de la fonction permettent d'obtenir le nombre de mois ou de jours résiduels une fois les années soustraites, ou les mois et années soustraites :

	B	C	D	H	I	J
	Chantier	Date de démarrage	Date de fin	Nombre de mois (années soustraites)	Nombre de jours (années soustraites)	Nombre de jours (années et mois soustraits)
29						
30	Chantier 1	13/05/2003	11/07/2008	1	59	28
31	Chantier 2	03/07/2008	04/08/2009	1	32	1
32	Chantier 3	02/07/2008	03/09/2008	2	63	1
33						

=DATEDIF(C32;D32;"ym")

=DATEDIF(C32;D32;"yd")

=DATEDIF(C32;D32;"md")

Toujours pour notre chantier 3, nous avons obtenu zéro pour le nombre d'années, 2 mois ou 63 jours, ces deux résultats ne changent donc pas ; en revanche, l'intervalle « md » restant, une fois les années et les mois retranchés, est bien de 1 jour : 63 moins 2 mois de 31 jours reste 1.

Idem pour le chantier 1, qui a duré 5 ans ou 61 mois ou 1886 jours. Le résultat de 1886 jours n'est pas forcément très parlant, on se représente mieux la durée exprimée en années mois et jours sur une longue durée. En fonction des intervalles choisis DATEDIF permet de dire que ce chantier a duré en tout 5 ans 1 mois et 28 jours.

Ce résultat peut être présenté dans un tableau comme dans notre exemple, mais il peut s'avérer utile de l'écrire en texte. Utilisons pour cela l'opérateur de concaténation : & (équivalent à la fonction concatener())

Pour obtenir les années : =DATEDIF(C30;D30;"y") renvoie 5; la formule
=DATEDIF(C30;D30;"y")&"ans" renvoie le texte 5 ans.

Le même raisonnement est applicables pour les mois et les jours. Pour écrire sous forme de texte le résultat 5 ans 1 mois 28 jours il suffit donc d'appliquer la formule suivante :

Formule :

```
=DATEDIF(C30;D30;"y")&" ans "&DATEDIF(C30;D30;"ym")&" mois "&DATEDIF(C30;D30;"md")&" jours"
```

Comment gérer le singulier/pluriel :

Appliquée au chantier 2 (soit en remplaçant dans la formule ci dessus la ligne 30 par la ligne 31) cette formule renvoie : **1 ans 1 mois 1 jours** . Le pluriel de ans et jours est ici mal venu.

- Afin que le résultat soit correct aussi bien au singulier qu'au pluriel, une solution serait de mettre les "s" entre parenthèses :

La formule `=DATEDIF(C31;D31;"y")&" an(s)"` renvoie le texte **1 an(s)**

- Une autre solution est de tester si le nombre d'années/jours sont supérieurs à 1:

Par exemple, pour le nombre d'années : `=SI(DATEDIF(C31;D31;"y")>1;" ans ";" an ")` renvoie ans au pluriel si le nombre d'années est >1, an au singulier dans le cas contraire.

Le raisonnement est le même pour le nombre de jours, et pour donner le résultat **1 an 1 mois 1 jour** la formule devient :

Formule :

```
=DATEDIF(C31;D31;"y") & SI(DATEDIF(C31;D31;"y")>1;" ans ";" an ") &  
DATEDIF(C31;D31;"ym") & " mois " & DATEDIF(C31;D31;"md") & SI(DATEDIF(  
C31;D31;"md")>1;" jours ";" jour ")
```

Comment gérer également les valeurs nulles :

Appliquée au chantier 3 notre dernière formule ci dessus donne: **0 an 2 mois 1 jour**. Le problème du singulier et du pluriel est résolu, mais ce sont maintenant les zéros qui sont inesthétiques.

Sur son excellent site, [Laurent Longre](#) (site aujourd'hui hébergé sur excelabo) met à notre disposition une formule qui, à l'aide d'un format personnalisé, de la fonction datedif() associée aux fonctions texte() et suppresspace(), permet de résoudre à la fois le problème d'orthographe et celui des valeurs nulles.

Formule de [Laurent Longre](#) adaptée à notre exemple :

Formule :

```
=SUPPRESPEACE(SI(D32-C32;TEXTE(DATEDIF(C32;D32;"y");">1]0"" ans"";[>]"1 an"";")  
&TEXTE(DATEDIF(C32;D32;"ym");">] 0"" mois "";") &TEXTE(DATEDIF(C32;D32;"md");
```

```
"[>1]0" jours";[>]"1 jour";");"0 jour"))
```

Superbe formule, applicable à tous les cas rencontrés, et qui renvoie bien 2 mois 1 jour dans notre exemple.

Cette formule est puissante, mais peut sembler complexe si l'on débute avec Excel; il peut y avoir d'autres approches, certes plus longues et moins élégantes, en passant par un tableau intermédiaire ce qui permet de n'utiliser que des formulations simples.

Par exemple dans le tableau ci dessous, trois formules pour les années/mois/jours gérant les valeurs nulles ainsi que le singulier/pluriel, et la dernière synthétisant l'ensemble :


	B	C	D	E	F	G	H
	Chantier	Date de démarrage	Date de fin	Nombre d'années	Nombre de mois (années soustraites)	Nombre de jours (années et mois soustraits)	Résultat en texte
1							
2	Chantier 1	13/05/2003	11/07/2008	5 ans	1 mois	28 jours	5 ans 1 mois 28 jours
3	Chantier 2	03/07/2008	04/08/2009	1 an	1 mois	1 jour	1 an 1 mois 1 jour
4	Chantier 3	02/07/2008	03/09/2008		2 mois	1 jour	2 mois 1 jour
5							

Pour les années en colonne E:

Formule :

```
=SI(DATEDIF(C4;D4;"y")=0;"";SI(DATEDIF(C4;D4;"y")=1;DATEDIF(C4;D4;"y")&" an";DATEDIF(C4;D4;"y")&" ans"))
```

Décomposons la formule :



- Si le résultat de DATEDIF(C4 ;D4) est égal à zéro, cela signifie qu'il y a moins d'un an entre les deux dates, dans ce cas nous n'affichons pas le résultat et mettons un vide (deux guillemets)
- Si non nous faisons un second test : Si le résultat de DATEDIF(C4 ;D4) est égal à 1, alors nous affichons le résultat suivi de an au singulier soit DATEDIF(C4 ;D4)& " an"
- Dernier cas possible une fois que les deux premiers ont été traités, il y a plusieurs années séparant les deux dates, nous affichons donc le résultat mais suivi de ans au pluriel soit en

formule DATEDIF(C4;D4)& " ans".

Pour les mois en colonne F:

Formule :

=SI (DATEDIF (C4 ; D4 ; "ym") = 0 ; " " ; DATEDIF (C4 ; D4 ; "ym") & " mois")

La formule est plus simple puisque mois est invariable, il suffit de tester si le nombre de mois est nul afin d'afficher soit vide (avec deux guillemets sans espace "") soit le résultat suivi de mois.

Pour les jours en colonne G:

Formule :

=SI (DATEDIF (C4 ; D4 ; "md") = 0 ; " " ; SI (DATEDIF (C4 ; D4 ; "md") = 1 ; DATEDIF (C4 ; D4 ; "md") & " jour" ; DATEDIF (C4 ; D4 ; "md") & " jours"))

Même principe que pour les années, un premier test pour gérer les valeurs nulles, suivi d'un second pour écrire jour au singulier ou au pluriel.

La concaténation colonne H:

Formule :

=SUPPRESPEACE (E4 & " " & F4 & " " & G4)

Le résultat des trois formules précédentes est agrégé et séparé par un espace afin que les résultats ne soient pas collés les uns aux autres (=SUPPRESPEACE(E4&F4&G4) donnerait 2 mois1 jour sans séparation entre mois et 1 ce qui est inesthétique).

Le tout est intégré dans la fonction SUPPRESPEACE() afin de supprimer les espaces inutiles. En effet, dans notre exemple n'ayant pas d'année complète la cellule E4 est vide puis suivie d'un espace avant d'inscrire le nombre de mois. Cet espace est inutile, la fonction SUPPRESPEACE() permet de l'éliminer.



A noter:

- Lorsque les dates sont saisies directement dans la formule, les noter entre guillemets:
Exemple : DATEDIF("13/03/1964";"31/12/2009";"y")
- DATEDIF() gère les années bissextiles.

Pour effectuer des calculs sur des âges, ancienneté du personnel, date de règlement des factures pour gérer des relances, il est souvent bien utile de connaître cette fonction DATEDIF.

En complément à cet article, pour voir d'autres exemples et vous permettre de vous entraîner, un fichier Excel est disponible en téléchargement. Il vous présente quelques exercices corrigés, adaptés à des situations concrètes :

Téléchargement

Extrait du fichier d'exercices :



Dépassement d'échéance

Dans cet exemple, nous disposons d'une liste de factures non payées, leur date d'émission ainsi que leur date d'échéance.

Nous cherchons à savoir si les dates d'échéance sont dépassées et de combien de jours.

Si le dépassement est supérieur à 30 jours, nous devons relancer le client.

Dans la colonne F calculer la durée de dépassement en nombre de jours au format nombre (exemple 12).

N° Facture	Client	Date Facture	Date du jour	Dépassement en jours	A relancer
19573	Lefebvre	05/07/2009	25/07/2009		
35987	Pucheral	04/06/2009	25/07/2009		
25789	Blanchard	18/06/2009	25/07/2009		
25254	André	15/07/2009	25/07/2009		
20124	Degas	27/06/2009	25/07/2009		



La solution et d'autres
exercices corrigés dans
le fichier attaché ...